Supplementary material

Isotope-filtered nD NMR spectroscopy to unravel the molecular structures of phenolic compounds in tagged soil organic matter.

N. G. A. Bell, M. C. Graham and D. Uhrin

Table of content,

3D HMQC-HMBC

;**hmqc_hmbc_3D**
;avance-version (07/04/04)
;gradient selected refocussed HMBC with preceding HMQC step for methyl group to ipso carbon
;correlations via 3 bond long range ^{1}H-^{13}C couplings
;phase sensitive using Echo/Antiecho gradient selection
;decoupling during acquisition

;$CLASS=HighRes
;$DIM=2D
;$TYPE=
;$SUBTYPE=
;$COMMENT=

#include <Avance.incl>
#include <Grad.incl>
#include <Delay.incl>

"cnst30=(1-sfo2/sfo1)/(1+sfo2/sfo1)"
define list<gradient> EA1 = { 1.000 -cnst30}
define list<gradient> EA2 = { -cnst30 1.000}
"p2=p1*2"
"p4=p3*2"
"d6=cnst12/(cnst2*2)"
"d2=1/2*cnst2"
"d12=20u"
"TAU=d2-d16-p16-(p1*2/3.1416)"
"TAU1=d2-d16-p16-p3"
"in0=inf1/2"
"in10=inf2/2"
"DELTA3=0.5*(d6-p25)"
"DELTA4=0.5*(d6-p25)-de"

ifdef F1180
"d0=0.5*in0"
"DELTA=p2"
"DELTA2=p16+d16+0.5*p24+p1"
else
"d0=3u"
"DELTA=p2+d0*2"
"DELTA2=p16+d16+0.5*p24+p1+d0"
endif

ifdef F2180
"d10=0.5*in10-p1-(p3*2/3.1416)"
else
"d10=3u"
endif

"acqt0=0"
baseopt_echo
1 ze

```
2 30m
  50u BLKGRAD
  d1 do:f2 pl2:f2
  50u UNBLKGRAD
  20u fq=cnst21(bf ppm):f2   ;cnst2=middle of the OMe 13C

  (p1 ph1):f1
  TAU
3 p16:gp1
  d16

  (p3 ph3):f2
  d10
  (p2 ph1):f1
  d10
  (p3 ph1):f2

  p16:gp1
  d16
  TAU1
  (p3 ph1):f2

  DELTA2
  DELTA3 pl0:f2
  4u
  (p25:sp8 ph1):f2
  4u
  DELTA3 pl2:f2
  DELTA2 fq=cnst22(bf ppm):f2   ;cnst22=o2p middle of methyl and aromatic carbons

  (p3 ph4):f2
  d0
  p2 ph2
  d0
  p16:gp2*EA1
  d16 pl0:f2
  (p24:sp7 ph5):f2
  DELTA
  p16:gp2*EA2
  d16 pl2:f2
  (p3 ph5):f2

  DELTA3 pl0:f2 fq=cnst21(bf ppm):f2   ;cnst21=middle of the OMe 13C
  4u
  (p25:sp8 ph1):f2
  4u
  DELTA4 pl12:f2
  go=2 ph31 cpds2:f2
  30m do:f2 mc #0 to 2
  F2PH(calph(ph3, 90), caldel(d10, +in10))

# ifdef F1SHIFT
    F1EA(calgrad(EA1) & calgrad(EA2), caldel(d0, +in0) & calph(ph31, +180) & calph(ph4, +180) &
calph(ph4, -115))
# else
```

```
    F1EA(calgrad(EA1) & calgrad(EA2), caldel(d0, +in0) & calph(ph4, +180) & calph(ph31, +180))
# endif

  50u BLKGRAD

exit

ph1=0
ph2=0 0 0 0 0 0 0 0
   2 2 2 2 2 2 2 2
ph3=0 0 2 2
ph4=0 2
ph5=0 0 0 0 2 2 2 2
ph6=0
ph31=0 2 2 0 2 0 0 2

;pl0 : f2 channel - power level for pulse
;pl2 : f2 channel - power level for pulse
;pl12 : f2 channel - power level for decoupling
;sp7 : f2 channel - shaped pulse (180degree refocussing)
;spnam7 : Crp60comp.4
;sp8 :  f2 channel - shaped pulse
;spnam8 : Rsnob.1000
;p1 : f1 channel -  90 degree high power pulse
;p2 : f1 channel - 180 degree high power pulse
;p3 : f2 channel -  90 degree high power pulse
;p16 : homospoil/gradient pulse                    [1 msec]
;p24 : f2 channel - 180 degree shaped pulse for refocussing
;     = 2msec for Crp60comp.4
;p25 : f2 methyl selective 150us Rsnob.1000
;d0 : incremented delay for aromatic carbons                    [3 usec]
;d1 : relaxation delay; 1-5 * T1
;d6 : delay for evolution of long range couplings (1/2Jlr)
;d10 : incremental delay for methyl carbons
;d16 : delay for homospoil/gradient recovery
;cnst2 : = JMe =144.5 Hz
;cnst21 : =  middle of the OMe 13C, e.g. 56.6 ppm
;cnst12 : = 24 optimum for 6Hz nJCC
;cnst22 : = o2p middle of the methyl and aromatic carbons
;inf1 : 1/SW(X) = 2 * DW(X)
;in0 : 1/(2 * SW(X)) = DW(X)
;nd0 : 2
;NS : 4 * n
;DS : 16
;td1 : number of experiments
;FnMODE : echo-antiecho
;cpdsf2 : decoupling

;for z-only gradients:
;gpz1 : 10%
;gpz2 : 80%

;use gradient files:
;gpnam1 : SMSQ10.100
;gpnam1 : SMSQ10.100
```

3D $HcCH_3$

;**HcCH3_3D**
;avance-version (02/05/31)

;$CLASS=HighRes
;$DIM=1D
;$TYPE=
;$SUBTYPE=
;$COMMENT=

#include <Avance.incl>
#include<Grad.incl>
#include<Delay.incl>

"p2=p1*2"
"p4=p3*2"
"d4=1s/(cnst1*4)" ; cnst1=^{1}JCHarom
"d11=30m"
"d12=20u"
"d3=0.0915s/cnst3" ; cnst3=$^{1}JCH_3$
"d22=(1s/cnst2*4)" ; cnst2=^{n}JCC
"d27=1s/(cnst4*4)" ; consider 1/cnst4*8 if two methoxy groups surrounding one aromatic proton.
"DELTA1=1s/(cnst3*4)-p16-d16+de-p1*0.64"
"DELTA2=1s/(cnst3*4)-p16-d16"
"in31=inf1/2"
"in33=inf2/2"
"in43=in33"
"d32=d22-d12-p16-d16"
"TAU=d22-p2-d4-p16-d16"
"TAU2=d27-p2-d3-p16-d16"

ifdef F1180
"d31=0.5*in31"
"TAU1=d4+p4"
else
"d31=3u"
"TAU1=d4+2*d31+p4"
endif

ifdef F2180
"d33=d27-d12-0.5*in33-p16-d16"
"d43=d3+0.5*in33"
else
"d33=d27-d12-p16-d16"
"d43=d3"
endif

ifdef REFOCUS
"acqt0=0"
baseopt_echo
else
"acqt0=(-2/PI)*p1"
endif

;aqseq 321 ; aqmode takes td1 as inner loop and td2 as outer loop counters

```
1 ze
  1m fq=cnst12(bf ppm):f2
2 30m do:f2
  50u BLKGRAD
  d1
  d12 pl1:f1 pl2:f2
  50u UNBLKGRAD

  50u fq=cnst23(bf ppm):f1
  d16 fq=cnst13(bf ppm):f2

  (p1 ph3):f1      ; this is the real start

  p19:gp1
  d16 pl0:f1
  (p12:sp2 ph4:r):f1
  p19:gp1
  d16

  p19:gp2
  d16
  (p12:sp2 ph8:r):f1
  p19:gp2
  d16 pl1:f1
  p28 ph6

  d31
  (p4 ph1):f2
  d31
  d4
  (center (p4 ph1):f2 (p2 ph1):f1)
  TAU1
  (p1 ph2):f1      ; zz state
  50u fq=cnst14(bf ppm):f2
  50u fq=cnst24(bf ppm):f1
  p16:gp7
  d16

  (p3 ph5):f2
  d12 pl0:f2
  d32
  p16:gp8
  d16
  4u
  (p24:sp7 ph1):f2
  4u
  p16:gp8
  d16
  TAU pl2:f2
  (p2 ph1):f1
  d4
```

```
 (p3 ph1):f2              ; t2 period
 d12 pl0:f2
 d33                      ; decrementing t2
 p16:gp3
 d16
 4u
 (p24:sp7 ph1):f2         ; shape moving
 4u
 p16:gp3
 d16
 TAU2 pl2:f2
 (p2 ph1):f1
 d43                      ; incrementing t2
 (p3 ph9):f2              ; refocuse nJCC

 p16:gp4
 d16
50u fq=cnst22(bf ppm):f1
50u fq=cnst12(bf ppm):f2

 (p1 ph1):f1

# ifdef REFOCUS
 DELTA1
 p16:gp5
 d16
 (center (p2 ph1):f1 (p4 ph1):f2)
 p16:gp5
 d16 pl12:f2
 DELTA2
 go=2 ph31 cpds2:f2
# else
 go=2 ph31
# endif


d11 do:f2 mc #0 to 2

# ifdef F2SHIFT
F2PH(calph(ph9, -90), caldel(d33, -in33) & caldel(d43, +in43) & calph(ph9, -140))
# else
F2PH(calph(ph9, -90), caldel(d33, -in33) & caldel(d43, +in43))
# endif

# ifdef F1SHIFT
F1PH(calph(ph3, +90) & calph(ph6, +90), caldel(d31, +in31) & calph(ph3,-20) & calph(ph4, -20) &
calph(ph6, -20) & calph(ph8, -20))
# else
F1PH(calph(ph3, +90) & calph(ph6, +90), caldel(d31, +in31))
# endif

50u BLKGRAD

exit
```

ph1=0
ph2=1
ph3=0
ph4={0}*4 {1}*4
ph6=1
ph8=0 0 1 1
ph9=0 2
ph5=0
ph10=1
ph31=0 2 2 0 2 0 0 2

;pl0 : f1 power level for shaped pulse
;pl1 : f1 power level for pulse (default)
;pl2 : f2 power level for pulse (default)
;pl12: f2 power level for CPD/BB decoupling
;p1 : f1 90 degree high power pulse
;p2 : f1 180 degree high power pulse
;p3 : f2 90 degree high power pulse
;p4 : f2 180 degree high power pulse
;p12 : shaped pulse Rsnob.1000
;p24 : shaped pulse crp60comp.4
;sp2 : shaped pulse Rsnob.1000
;sp28 : trim pulse
;sp7 : shaped pulse crp60comp.4
;p16 : gradient pulse
;d16 : gradient pulse delay
;d1 : relaxation delay; 1-5 * T1
;d3 : 0.0915s/cnst3
;d4 : 1s/(cnst1*4)
;d11 : 30m
;d12 : 20u;
;d3 : 0.0915s/cnst3
;d22 : (1s/cnst2*4)
;d27 : 1s/(cnst2*4)
;d32 : d22-d12-p16-d16
;d33 : incrementing t1
;d43 : incrementing t2
;DELTA : d27-p2-d3
;DELTA1 : 1s/(cnst3*4)+de-p1*0.78
;DELTA2 : 1s/(cnst3*4)-p16-d16
;cnst1 : ^{1}JCHarom
;cnst2 : ^{3}JCC
;cnst3 : ^{1}JCHmet
;cnst12 : ^{13}Cmet
;cnst13 : ^{13}Carom
;cnst14 : ^{13}Cmet and arom
;cnst22 : ^{1}Hmet
;cnst23 : ^{1}Harom
;cnst24 : ^{1}Hmet and arom
;cpd2 : decoupling according to sequence defined by cpdprg2
;pcpd2 : f2 channel - 90 degree pulse for decoupling sequence
;NS : 4 * n
;DS : 4

;td1 : number of experiments
;FnMODE : States-TPPI

;for z-only gradients:
;gpz1 : 11%
;gpz2 : 50%
;gpz3 : 23%
;gpz4 : 8%
;gpz5 : 17%
;gpz6 : 33%
;gpz7 : 10%

;use gradient files:

;gpnam1 : SMSQ10.100
;gpnam2 : SMSQ10.100
;gpnam3 : SMSQ10.100
;gpnam4 : SMSQ10.100
;gpnam5 : SMSQ10.100
;gpnam6 : SMSQ10.100

3D $hCCH_3$

;**hCCH3_3D**
;avance-version (02/05/31)

;$CLASS=HighRes
;$DIM=1D
;$TYPE=
;$SUBTYPE=
;$COMMENT=

#include <Avance.incl>
#include<Grad.incl>
#include<Delay.incl>

"p2=p1*2"
"p4=p3*2"
"d4=1s/(cnst1*4)" ; cnst1=^{1}JCHarom
"d11=30m"
"d12=20u"
"d3=0.0915s/cnst3" ; cnst3=$^{1}JCH_3$
"d22=(1s/cnst2*4)" ; cnst2=^{n}JCC
"d27=1s/(cnst4*4)"
"DELTA1=1s/(cnst3*4)-p16-d16+de-p1*0.64"
"DELTA2=1s/(cnst3*4)-p16-d16"
"in32=inf1/2"
"in33=inf2/2"
"in42=in32"
"in43=in33"
"TAU=1s/(cnst2*4)-p2-d4-p16-d16"
"TAU2=d27-p2-d3-p16-d16"

ifdef F1180
"d32=d22-d12-0.5*in32-p16-d16"
"d42=d4+0.5*in32"
else
"d42=d4"
"d32=d22-d12-p16-d16"
endif

ifdef F2180
"d33=d27-d12-0.5*in33-p16-d16"
"d43=d3+0.5*in33"
else
"d33=d27-d12-p16-d16"
"d43=d3"
endif

ifdef REFOCUS
"acqt0=0"
baseopt_echo
else
"acqt0=(-2/PI)*p1"

```
# endif

;aqseq 321 ; aqmode takes  td1 as inner loop and td2 as outer loop counters


1 ze
  1m fq=cnst12(bf ppm):f2
2 30m do:f2
  50u BLKGRAD
  d1
  d12 pl1:f1 pl2:f2
  50u UNBLKGRAD

  50u fq=cnst23(bf ppm):f1
  d16 fq=cnst13(bf ppm):f2

  (p1 ph3):f1      ; this is the real start

  p19:gp1
  d16 pl0:f1
  (p12:sp2 ph4:r):f1
  p19:gp1
  d16

  p19:gp2
  d16
  (p12:sp2 ph8:r):f1
  p19:gp2
  d16 pl1:f1
  p28 ph6

  d4
  (center (p4 ph1):f2 (p2 ph1):f1)
  d4
  (p1 ph2):f1      ; zz state
  50u fq=cnst14(bf ppm):f2
  50u fq=cnst24(bf ppm):f1
  p16:gp3
  d16

  (p3 ph5):f2                ; t1 period
  d12 pl0:f2
  d32                        ; decrementing t1
  p16:gp4
  d16
  4u
  (p24:sp7 ph1):f2           ; shape moving
  4u
  p16:gp4
  d16
  TAU pl2:f2
  (p2 ph1):f1
  d42                        ; incrementing t1

  (p3 ph1):f2               ; t2 period
```

```
  d12 pl0:f2
  d33                    ; decrementing t2
  p16:gp5
  d16
  4u
  (p24:sp7 ph1):f2       ; shape moving
  4u
  p16:gp5
  d16
  TAU2 pl2:f2
  (p2 ph1):f1
  d43                ; incrementing t2
  (p3 ph9):f2        ; refocuse nJCC

  p16:gp6
  d16
 50u fq=cnst22(bf ppm):f1
 50u fq=cnst12(bf ppm):f2

  (p1 ph1):f1

# ifdef REFOCUS
  DELTA1
  p16:gp7
  d16
  (center (p2 ph1):f1 (p4 ph1):f2)
  p16:gp7
  d16 pl12:f2
  DELTA2
  go=2 ph31 cpds2:f2
# else
  go=2 ph31
# endif

  50u BLKGRAD

 d11 do:f2 mc #0 to 2

# ifdef F2SHIFT
 F2PH(calph(ph9, -90), caldel(d33, -in33) & caldel(d43, +in43) & calph(ph9, -140))
# else
 F2PH(calph(ph9, -90), caldel(d33, -in33) & caldel(d43, +in43))
# endif

# ifdef F1SHIFT
 F1PH(calph(ph5, -90), caldel(d32, -in32) & caldel(d42, +in42) & calph(ph5, +165))
# else
 F1PH(calph(ph5, -90), caldel(d32, -in32) & caldel(d42, +in42))
# endif

exit

ph1=0
ph2=1
ph3=0
```

```
ph4={0}*4 {1}*4
ph5=0 2
ph6=1
ph8=0 0 1 1
ph9=0
ph10=1
ph31=0 2 2 0 2 0 0 2


;pl0 : f1 power level for shaped pulse
;pl1 : f1 power level for pulse (default)
;pl2 : f2 power level for pulse (default)
;pl12: f2 power level for CPD/BB decoupling
;p1 : f1 90 degree high power pulse
;p2 : f1 180 degree high power pulse
;p3 : f2 90 degree high power pulse
;p4 : f2 180 degree high power pulse
;p12 : shaped pulse Rsnob.1000
;p24 : shaped pulse crp60comp.4
;sp2 : shaped pulse Rsnob.1000
;sp28 : trim pulse
;sp7 : shaped pulse crp60comp.4
;p16 : gradient pulse
;d16 : gradient pulse delay
;d1 : relaxation delay; 1-5 * T1
;d3 : 0.0915s/cnst3
;d4 : 1s/(cnst1*4)
;d11 : 30m
;d12 : 20u;
;d3 : 0.0915s/cnst3
;d22 : (1s/cnst2*4)
;d27 : 1s/(cnst2*4)
;d32 : decrementing t1
;d33 : incrementing t1
;d42 : decrementing t2
;d43 : incrementing t2
;DELTA  : d27-p2-d3
;DELTA1 ; 1s/(cnst3*4)+de-p1*0.78
;DELTA2 ; 1s/(cnst3*4)-p16-d16
;cnst1 : [1]JCHarom
;cnst2 : [3]JCC
;cnst3 : [1]JCHmet
;cnst12 : [13]Cmet
;cnst13 : [13]Carom
;cnst14 : [13]Cmet and arom
;cnst22 : [1]Hmet
;cnst23 : [1]Harom
;cnst24 : [1]Hmet and arom
;cpd2 : decoupling according to sequence defined by cpdprg2
;pcpd2 : f2 channel - 90 degree pulse for decoupling sequence
;NS : 4 * n
;DS : 4
;td1 : number of experiments
;FnMODE : States-TPPI
```

;for z-only gradients:
;gpz1 : 11%
;gpz2 : 50%
;gpz3 : 23%
;gpz4 : 8%
;gpz5 : 17%
;gpz6 : 33%
;gpz7 : 10%

;use gradient files:

;gpnam1 : SMSQ10.100
;gpnam2 : SMSQ10.100
;gpnam3 : SMSQ10.100
;gpnam4 : SMSQ10.100
;gpnam5 : SMSQ10.100
;gpnam6 : SMSQ10.100

3D HMQC-NOESY

```
;HMQC_NOESY_3D
;avance-version (09/04/15)
;3D H-1/X correlation via heteronuclear zero and double quantum
;coherence followed by NOESY
;phase sensitive
;with 180 pulse during mixing time
;
;$CLASS=HighRes
;$DIM=3D
;$TYPE=
;$SUBTYPE=
;$COMMENT=

#include <Avance.incl>
#include <Delay.incl>
#include <Grad.incl>

"p2=p1*2"
"p4=p3*2"
"d2=1s/(cnst2*2)"
"d11=30m"
"d12=20u"
"d13=4u"
"DELTA=d8*0.5-p16-d16"
"DELTA2=d8*cnst1-p16-d16"  ; cnst1 controls the spacing of 2 180 deg pulses during the mixing time, typically 0.28-0.4
"DELTA3=d8*(0.5-cnst1)-p16-d16"
"TAU=d2-d16-p16"
"in0=inf1/2"
"TAU1=6u+p4"
"in10=inf2/2"

# ifdef F1180
"d0=in0*0.5-p3"
# else
"d0=3u"
# endif

# ifdef F2180
"d10=in10*0.5-p1-(2/PI)*p3"
# else
"d10=3u"
# endif

"acqt0=-p1*2/3.1416"


1 ze
2 d11 do:f2
3 20u pl1:f1 BLKGRAD
  d1
```

```
  50u UNBLKGRAD
  p1 ph1
  p19:gp1
  d16 pl0:f1
  p12:sp2:f1 ph12:r
  p19:gp1
  d16
  p19:gp2
  d16
  p12:sp2:f1 ph11:r
  p19:gp2
  d16 pl1:f1  ; end of dpfgse
  TAU

# ifdef F1180
# else
  TAU1
# endif

  p16:gp3
  d16
  p3:f2 ph3
  d10       ; t2 period
  p2 ph2
  d10
  p3:f2 ph4
  p16:gp3
  d16
  TAU
  d0      ; t1 period
  p4:f2 ph7
  d0
  p1 ph6

# ifdef MIX180
  p16:gp4
  d16 pl0:f1
  DELTA             ; d8*0.5-p16-d16
  (p11:sp1 ph11):f1   ; shaped pulse during mixing time
  DELTA
  p16:gp4*-1
  d16  pl1:f1
# endif

# ifdef MIX2180
  p16:gp4
  d16 pl0:f1
  DELTA2            ; d8*0.5-p16-d16
  (p11:sp1 ph11):f1   ; shaped pulse during mixing time
  DELTA3
  p16:gp4*-1
  d16

  p16:gp5
  d16
```

```
  DELTA2              ; d8*0.5-p16-d16
  (p11:sp1 ph12):f1   ; shaped pulse during mixing time
  DELTA3
  p16:gp5*-1
  d16 pl1:f1
# endif

# ifdef NO180
  d8
# endif

  d12
  p1 ph5
  go=2 ph31 ;cpd2:f2
  d11 do:f2 mc #0 to 2

F1PH(calph(ph1, -90), caldel(d0, +in0))
F2PH(calph(ph3, +90), caldel(d10, +in10))

 50u BLKGRAD
exit

 ph1=0 0 2 2
 ph2=0
 ph3=0 2
 ph4=0
 ph5=0 0 0 0 2 2 2 2
 ph6=0
 ph7=0
 ph11=1
 ph12=0
 ph31=0 2 2 0 2 0 0 2


;pl0 : f1 - power level for pulse
;pl1 : f1 - power level for pulse (default)
;pl2 : f2 - power level for pulse
;pl9 : f1 - power level for presaturation
;pl12: f2 - power level for CPD/BB decoupling
;p1 : f1 -  90 degree high power pulse
;p2 : f1 - 180 degree high power pulse
;p3 : f2 -  90 degree high power pulse
;p16 : 1ms gradient pulse
;p19 : 600us gradient pulse
;p11 : shaped pulse during mixing time IBURP
;p12 : DPFGSE Rsnob.1000
;sp1 : shaped pulse during mixing time IBURP
;sp2 : DPFGSE Rsnob.1000
;d0 : incremented period                [3 usec]
;d10 : incremented period
;d1 : relaxation delay; 1-5 * T1
;d2 : 1/(2J)XH
;d8 : mixing time
;d11 : delay for disk I/O                    [30 msec]
;d12 : delay for power switching                [20 usec]
```

;d13 : short delay
;d16 : gradient pulse
;cnst2 : J(XH)
;cnst1 : d8 180 degree pulse spacing 0.28-0.4
;inf1 : 1/SW(X) = 2 * DW(X)
;in0 : 1/(2 * SW(X)) = DW(X)
;nd0 : 2
;NS : 16 * n
;DS : 16
;td1: number of experiments
;FnMODE : States-TPPI, TPPI, States or QSEQ
;cpd2 : decoupling according to sequence defined by cpdprg2
;pcpd2 : f2 channel - 90 degree pulse for decoupling sequence

;for z-only gradients:
;gpz1 : 33%
;gpz2 : 17%
;gpz3 : 12%
;gpz4 : 9%
;gpz5 : 25%

;use gradient files:

;gpnam1 : SMSQ10.50
;gpnam2 : SMSQ10.50
;gpnam3 : SMSQ10.100
;gpnam4 : SMSQ10.100
;gpnam5 : SMSQ10.100

3D HMQC-NOESY-TOCSY

;**HMQC_NOESY_TOCSY_3D**
;avance-version (09/04/15)
;3D H-1/X correlation via heteronuclear zero and double quantum
;coherence followed by NOESY and TOCSY steps
;phase sensitive
;with decoupling during acquisition
;with 180 pulse during mixing and dpfgse
;
;$CLASS=HighRes
;$DIM=3D
;$TYPE=
;$SUBTYPE=
;$COMMENT=

#include <Avance.incl>
#include <Delay.incl>
#include <Grad.incl>

"FACTOR1=(d9/(p6*115.112))/2"
"l1=FACTOR1*2"
"p4=2*p3"
"p2=p1*2"
"d2=1s/(cnst2*2)"
"d11=30m"
"d12=20u"
"d13=4u"
"DELTA=d8*0.5-p16-d16"
"DELTA2=d8*cnst1-p16-d16" ; cnst1 controls the spacing of 2 180 deg pulses during mixing time, typically 0.28-0.4
"DELTA3=d8*(0.5-cnst1)-p16-d16"
"TAU=d2-d16-p16"
"in0=inf1/2"
"TAU1=6u+p4"
"in10=inf2/2"

ifdef F1180
"d0=in0*0.5-p3"
else
"d0=3u"
endif

ifdef F2180
"d10=in10*0.5-p1-(2/PI)*p3"
else
"d10=3u"
endif
"acqt0=-p1*2/3.1416"

1 ze

```
2 d11 do:f2
3 20u pl1:f1 BLKGRAD
  d1
  50u UNBLKGRAD
  p1 ph1
  p19:gp1
  d16 pl0:f1
  p12:sp2:f1 ph12:r
  p19:gp1
  d16
  p19:gp2
  d16
  p12:sp2:f1 ph11:r
  p19:gp2
  d16 pl1:f1  ; end of dpfgse
  TAU

# ifdef F1180
# else
  TAU1
# endif

  p16:gp3
  d16
  p3:f2 ph3
  d10       ; t2 period
  p2 ph2
  d10
  p3:f2 ph4
  p16:gp3
  d16
  TAU
  d0      ; t1 period
  p4:f2 ph7
  d0
  p1 ph6

# ifdef HARD2180
  p16:gp4
  d16
  DELTA2            ; d8*0.5-p16-d16
  (p2 ph4):f1   ; hard pulse during mixing time
  DELTA3
  p16:gp4*-1
  d16

  p16:gp5
  d16
  DELTA2            ; d8*0.5-p16-d16
  (p2 ph4):f1   ; hard pulse during mixing time
  DELTA3
  p16:gp5*-1
  d16
# endif
```

```
# ifdef MIX180
  p16:gp4
  d16 pl0:f1
  DELTA             ; d8*0.5-p16-d16
  (p11:sp1 ph11):f1   ; shaped pulse during mixing time
  DELTA
  p16:gp4*-1
  d16  pl1:f1
# endif

# ifdef MIX2180
  p16:gp4
  d16 pl0:f1
  DELTA2            ; d8*0.5-p16-d16
  (p11:sp1 ph11):f1   ; shaped pulse during mixing time
  DELTA3
  p16:gp4*-1
  d16

  p16:gp5
  d16
  DELTA2            ; d8*0.5-p16-d16
  (p11:sp1 ph12):f1   ; shaped pulse during mixing time
  DELTA3
  p16:gp5*-1
  d16 pl1:f1
# endif

  p16:gp6
; 10u gron0
; (p32:sp29 ph1):f1
; 20u groff
   d16 pl10:f1

                                    ; begin DIPSI2
4 p6*3.556 ph23
  p6*4.556 ph25
  p6*3.222 ph23
  p6*3.167 ph25
  p6*0.333 ph23
  p6*2.722 ph25
  p6*4.167 ph23
  p6*2.944 ph25
  p6*4.111 ph23

  p6*3.556 ph25
  p6*4.556 ph23
  p6*3.222 ph25
  p6*3.167 ph23
  p6*0.333 ph25
  p6*2.722 ph23
  p6*4.167 ph25
  p6*2.944 ph23
  p6*4.111 ph25
```

```
  p6*3.556 ph25
  p6*4.556 ph23
  p6*3.222 ph25
  p6*3.167 ph23
  p6*0.333 ph25
  p6*2.722 ph23
  p6*4.167 ph25
  p6*2.944 ph23
  p6*4.111 ph25

  p6*3.556 ph23
  p6*4.556 ph25
  p6*3.222 ph23
  p6*3.167 ph25
  p6*0.333 ph23
  p6*2.722 ph25
  p6*4.167 ph23
  p6*2.944 ph25
  p6*4.111 ph23
  lo to 4 times l1

  p16:gp3
  d16
  10u gron0*1.333
  (p32*0.75:sp29 ph2):f1
  20u groff

  d16 pl1:f1

  p1 ph5
  go=2 ph31 ;cpd2:f2
  d11 do:f2 mc #0 to 2

F1PH(calph(ph1, -90), caldel(d0, +in0))
F2PH(calph(ph3, +90), caldel(d10, +in10))

  50u BLKGRAD
exit

 ph1=0 0 2 2
 ph2=0
 ph3=0 2
 ph4=0
 ph5=0 0 0 0 2 2 2 2
 ph6=0
 ph7=0
 ph11=1
 ph12=0
 ph23=3
 ph25=1
 ph31=0 2 2 0 2 0 0 2


;pl0 : f1 - power level for pulse
;pl1 : f1  - power level for pulse (default)
```

```
;pl2 : f2  - power level for pulse (default)
;pl12: f2  - power level for CPD/BB decoupling
;p1 : f1  -  90 degree high power pulse
;p2 : f1  - 180 degree high power pulse
;p3 : f2  -  90 degree high power pulse
;p6 : f1 - 90 degree low power pulse
;p16 : gradient pulse
;pl9 : gradient pulse
;p11 : shaped pulse during mixing time IBURP
;p12 : DPFGSE Rsnob.1000
;sp1 : shaped pulse during mixing time IBURP
;sp2 : DPFGSE Rsnob.1000
;d0 : incremented delay
;d10 : incremented delay              [3 usec]
;d1 : relaxation delay; 1-5 * T1
;d2 : 1/(2J)XH
;d8 : mixing time
;d9 : TOCSY mixing time
;d11 : delay for disk I/O                  [30 msec]
;d12 : delay for power switching                [20 usec]
;d13 : short delay
;d16 : delay for gradient recovery
;l1 : loop for DIPSI cycle: ((p6*115.112) * l1) = mixing time
;cnst1 : d8 180 degree pulse spacing 0.28-0.4                          [4 usec]
;cnst2 : = J(XH)
;inf1 : 1/SW(X) = 2 * DW(X)
;in0 : 1/(2 * SW(X)) = DW(X)
;nd0 : 2
;NS : 16 * n
;DS : 16
;td1 : number of experiments
;FnMODE : States-TPPI, TPPI, States or QSEQ
;cpd2 : decoupling according to sequence defined by cpdprg2
;pcpd2 : f2 channel - 90 degree pulse for decoupling sequence

;for z-only gradients:
;gpz1 : 33%
;gpz2 : 17%
;gpz3 : 12%
;gpz4 : 9%
;gpz5 : 25%
;gpz6 : 15%

;use gradient files:

;gpnam1 : SMSQ10.50
;gpnam2 : SMSQ10.50
;gpnam3 : SMSQ10.100
;gpnam4 : SMSQ10.100
;gpnam5 : SMSQ10.100
;gpnam6 : SMSQ10.100
```

Molecular structures obtained from the analysis of methylated fulvic acid isolated from a peat sample.

Fig. 1S. The methoxy region of the 2D ^{1}H , ^{13}C HSQC spectrum of methylated FA shown as 2D contour plot (left) and an intensity plot (right). The structures behind the most intense cross peaks 1-33 are shown below with corresponding identifiers. The nuclei for which the chemical shifts were obtained from the analysis of $^{13}CH_3O$- filtered nD NMR experiments are highlighted. The colour coding is as follows: green: methyl proton, black methyl carbon, red: aromatic carbons ipso or ortho or meta to the methoxy group, purple: aromatic protons ortho to the methoxy group, pink: aromatic protons meta to a methoxy group and turquoise: carbons obtained via carbonyl group correlations.



R_1, R_2: C(not O)
R_3: COCH=CHR



R_1, R_2, R_4: C (not O)
R_3: COCH=CHR



R_1, R_2, R_4: C (not O)
R_3: COCH=CHR



R_1, R_3: electron donating
R_2 : C,H
R_4 : electron withdrawing







R_1: CONHR, $COOCH_3$, $COCH_3$



R_1: COR, CH_3, Ph
R_2: C,H
R_3: C
R_4: electron withdrawing



R_1: C (OR if R_3: C)
R_2: C,H
R_3: OR (C if R_1: OR)
R_4: electron withdrawing





R_1:NH_2, NHR, NR_2



R_1: C
R_2: COCH=CHR
R_3: COR



R_2: COCH=CHR, COOR, COR









R_2: H, R
R_3: COR



R_1: OR, OAc
R_2: C(not O)



R_1: OR, OAc
R_2: C(not O)





R:CH_2OR, $CH{=}CH_2OAc$ (cis)















R: $CH(OR)_2$, CH_2OAc, $CH{=}CH{-}CH_2OAc$



R_1: H, R_2: COOR
R_1: CH_3, R_2: OR



R: OR, COOR



R: $CH(OR)_2$, CH_2OAc, $CH{=}CH{-}CH_2OAc$